# PISPIGOT - Producing Digits of $\pi$ one at a time 

© 2019 Valentín Albillo

Abstract<br>PISPIGOT is a proof-of-concept BASIC program written in 1996 for the HP-71B pocket computer to produce an arbitrary number of digits of $\pi$ one at a time using a so-called spigot algorithm.

Keywords: spigot algorithm, Pi, $\pi$, one digit at a time, multiprecision, HP-71B, pocket computer, BASIC, Hewlett-Packard

## 1. Introduction

PISPIGOT is a very short (6 lines, 285 bytes) BASIC program that I wrote in 1996 for the $H P-71 B$ pocket computer as a proof-of-concept example of implementing a spigot algorithm to produce an arbitrary number of digits of $\pi$ one at a time. It will also run in other HP BASIC models with minimal changes (see Note 1 below).

The algorithm produces N digits of $\pi$ (up to many thousands, limited only by available memory and time) one by one using just integer arithmetic on reasonably small integer values. None of the previous digits are needed once computed and no floating-point operations are needed at all whether single-precision or high-precision.

On the other hand, the number N of digits needs to be specified in advance (so it's not possible to add more digits to the previously produced ones without having to restart the whole computation) and an array with some $10 \mathrm{~N} / 3$ integer elements must be dimensioned to generate N digits, which is significantly more memory than needed by other multiprecision algorithms and also much slower. Thus, this algorithm and the resulting program featured here aren't intended to be competitive and are best considered as a proof-of-concept example.

The spigot algorithm is explained in detail in the reference given below. Basically, we start from the series:

$$
\frac{\pi}{2}=\sum_{i=0}^{\infty} \frac{i!}{(2 \mathrm{i}+1)!!}=1+\frac{1}{3}+\frac{1.2}{3.5}+\frac{1 \cdot 2.3}{3.5 \cdot 7}+\ldots=1+\frac{1}{3}\left(1+\frac{2}{5}\left(1+\frac{3}{7}\left(1+\frac{4}{9}(1+\cdots)\right)\right)\right)
$$

which can be considered a mixed-radix base $(1 / 3,2 / 5,3 / 7,4 / 9, \ldots)$ representation for $\pi / 2$, so the digits of $\pi$ itself in this base would all be 2 . The workings of the algorithm can be seen in the table below, taken from the reference. The array is initialized with the digits of $\pi$ in the mixed-radix base (all 2 ) and then the operations described in the leftmost column are performed, starting from the rightmost column and going left, column by column and row by row, from top to bottom, producing one digit of $\pi$ per row $(\mathbf{3}, \mathbf{1}, \mathbf{4}, \mathbf{1})$. We needed $[10 * \mathbf{4} / 3]=$ 13 columns (array elements) to get $\mathbf{4}$ digits.

|  | Digits of $\pi$ |  | $\frac{1}{3}$ | $\frac{2}{5}$ | $\frac{3}{7}$ | $\frac{4}{9}$ | $\frac{5}{11}$ | $\frac{6}{13}$ | $\frac{7}{15}$ | $\frac{8}{17}$ | $\frac{9}{19}$ | $\frac{10}{21}$ | $\frac{11}{23}$ | $\frac{12}{25}$ |
| :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: |
| Initialize |  | 2 | 2 | 2 | 2 | 2 | 2 | 2 | 2 | 2 | 2 | 2 | 2 | 2 |
|  |  | 20 | 20 | 20 | 20 | 20 | -20 | 20 | 20 | 20 | 20 | 20 | 20 | 20 |
| Carry |  |  |  |  |  |  |  |  |  |  |  |  |  |  |
| Remainders |  |  |  |  |  |  |  |  |  |  |  |  |  |  |
| $\times 10$ |  | 0 | 20 | 20 | 40 | 30 | 100 | 10 | 130 | 120 | 10 | 200 | 200 | 200 |
| Carry | $1<+13+\frac{20}{10}+\frac{33}{53}+\frac{40}{80}+\frac{65}{95}+48+98+88+72+\frac{150}{160}+\frac{132}{132}\left(\frac{96}{296}\right)=$ |  |  |  |  |  |  |  |  |  |  |  |  |  |
|  | $>13$ |  | 40 | 53 | 80 | 95 | 148 | 108 | 218 | 192 | 160 | 332 | 296 (200 |  |
| Remainders |  | $\pm 3$ | 1 | 3 | 3 | 5 | 5 | 4 | 8 | 5 | 8 | 17 |  | - |
|  |  | 30 | 10 | 30 | 30 | 50 | 50 | 40 | 80 | 50 | 80 | 170 | 200 | 0 |
| Carry |  |  | +24 | +30 | +40 | + 40 | + 42 | +63 | $+64$ | $+\underline{90}+$ | $+120$ | + 88 | +0 | = |
|  |  |  | 34 | 60 | 70 | 90 | 92 | 103 | 144 | 140 | 200 | 258 | 200 | 0 |
| Remainders |  | 1 | 1 | 0 | 0 | 0 | 4 | 12 | 9 | 4 | 10 | 6 | 16 | 0 |
|  |  | 10 | 10 | 0 | 0 | 0 | 40 | 120 | 90 | 40 | 100 | 60 | 160 | 0 |
| Carry | $1 \kappa$ |  | +2 | +9 | $+24$ | + 55 | +84 | +63 | +48 | +72 | +60 | +66 | $\underline{+0}$ | = |
|  |  |  | 12 | 9 | 24 | 55 | 124 | 183 | 138 | 112 | 160 | 126 | 160 | 0 |

When computing $\pi$ with this program these are the 5 last digits, RAM used and running times for $N$ digits:

| \# Digits | $\mathbf{1 0}$ | $\mathbf{2 0}$ | $\mathbf{4 0}$ | $\mathbf{5 0}$ | $\mathbf{1 0 0}$ | $\mathbf{2 0 0}$ | $\mathbf{5 0 0}$ | $\mathbf{8 0 0}$ | $\mathbf{1 0 0 0}$ |
| :--- | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: |
| 5 last digits | 92653 | 32384 | 84197 | $9375 \underline{0}^{*}$ | 17067 | 30381 | 19491 | 63185 | 20198 |
| $\sim$ RAM (bytes) | 280 | 380 | 580 | 680 | 1,180 | 2,180 | 5,180 | 8,180 | 10,180 |
| Time (seconds) | 0.19 | 0.75 | 3.01 | 4.67 | 18.73 | 75.14 | 472 | 1,212 | 1,896 |

(all running times in this paper are for the go71b emulator running at $128 x$ on a mid-range Android Samsung tablet)

The amount of $R A M$ needed is approximately $\mathbf{1 0 *} \mathbf{N} \mathbf{+ 1 8 0}$ bytes where $\mathbf{N}$ is the number of digits to compute so, for example, having 64 Kb of $R A M$ available would allow the computation of up to $\sim 6,500$ digits.

## 2. Program Listing

```
1 DESTROY ALL @ Z$="0000000000" @ T$="99999999999" @ INPUT N @ L=10*N DIV 3 @ INTEGER A(L)
2 FOR I=1 TO L @ A(I)=2 @ NEXT I @ M=0 @ P=0 @ FOR J=1 TO N @ Q=0 @ K=2*L+1
3 FOR I=L TO 1 STEP -1 @ K=K-2 @ X=10*A(I)+Q*I @ A(I)=MOD(X,K) @ Q=X DIV K @ NEXT I
4 A(1)=MOD (Q,10) @ Q=Q DIV 10 @ IF Q=9 THEN M=M+1 @ GOTO 6
5 IF Q#10 THEN PRINT STR$(P);T$[1,M]; @ P=Q @ M=0 ELSE PRINT STR$(P+1);Z$[1,M]; @ P=0 @ M=0
6 NEXT J @ PRINT STR$(P)
```


## 3. Usage Instructions

See the worked examples below to understand how to use the program.

## 4. Examples

The following examples can be useful to check that the program is correctly entered and to understand its usage.

### 4.1 Example

Produce the first 24 digits of $\pi$, then the first 76 digits.

### 4.2 Example

Produce the first $\mathbf{1 , 0 0 0}$ digits of $\pi$.

## RUN ? 1000 END LINE <br> 0


#### Abstract

3141592653589793238462643383279502884197169399375105820974944592307816406286208998628034825342117067 9821480865132823066470938446095505822317253594081284811174502841027019385211055596446229489549303819 6442881097566593344612847564823378678316527120190914564856692346034861045432664821339360726024914127 3724587006606315588174881520920962829254091715364367892590360011330530548820466521384146951941511609 4330572703657595919530921861173819326117931051185480744623799627495673518857527248912279381830119491 2983367336244065664308602139494639522473719070217986094370277053921717629317675238467481846766940513 2000568127145263560827785771342757789609173637178721468440901224953430146549585371050792279689258923 54201995611212902196086403441815981362977477130996 95024459455346908302642522308253344685035261931188 35982534904287554687311595628638823537875937519577 05187072113499999983729780499510597317328160963185 17101000313783875288658753320838142061717766914730 81857780532171226806613001927876611195909216420198


(all digits are $\mathrm{Ok}, \quad \sim 31^{\prime}$ )

## Notes

1. This program will run in several other HP BASIC computers by simply removing $H P 71 B$-specific statements such as DESTROY ALL. If the particular BASIC version does not allow for dinamic (re)dimensioning, then array A must be dimensioned to have $\operatorname{INT}(10 * N / 3)$ elements, where $N$ is the number of digits to compute. Furthermore, all variables used in the program (besides the $\mathbf{A}$ array) can be declared as integer for maximizing speed on some versions.
2. [*] Any digit-producing algorithm for normal numbers (which $\pi$ probably is) has the (unlikely) problem that the last digit(s) may be incorrect if there's a terminating string of 9 s , and even with no 9 s the very last digit might be wrong, as when computing $\mathrm{N}=50$ digits, where the last computed digit printed (underlined above) is a $\underline{\mathbf{0}}$ but should actually be a $\underline{\mathbf{1}}$.
3. I also wrote a version for the SHARP PC-1350/PC-1360 pocket computers and compatibles, see References.

## References

S. Rabinowitz and S. Wagonrancis (1995). A Spigot Algorithm for the Digits of $\pi$. The American Mathematical Monthly, Vol. 102, No. 3

Valentin Albillo (2019) SHARP Program VA135-PC-1350 Producing Digits of Pi one at a time

## Copyrights

Copyright for this paper and its contents is retained by the author. Permission to use it for non-profit purposes is granted as long as the contents aren't modified in any way and the copyright is acknowledged.

